


## Using Microsoft Excel

### Calculations in Excel

Although calculations are one of the main uses for spreadsheets, Excel can do most of the hard work for you by using a formula. When you enter a formula into a cell in a spreadsheet, Excel will calculate the answer and display the answer in that cell. There are a few rules to remember when creating a spreadsheet formula.




1. A formula must always begin with an equal sign (they will also work with a + or – sign but an equal sign is preferred). As soon as you begin a cell with a = Excel will know that you are creating a formula.
2. A formula will follow the order of operations (BIMDAS). Generally if there is more than one part to a formula the calculations will work from left to right but... Any part of the formula in brackets will be calculated first. Indices will be left. Multiplication and division will be next. Addition and subtraction will be calculated last.

B	Brackets	( )
I	Indices	^
M	Multiplication	*
D	Division	/
A	Addition	+
S	Subtraction	-

3. A formula can refer to other cells in the spreadsheet using cell references.
4. If any part of the formula is referring to text, the text must be enclosed in quotation marks “ ”.
5. The cells referred to in a formula can't include the cell the answer will be in. This will cause a circular reference error.

Each of these rules will be illustrated in the exercises that will follow.

**Exercise 1. Creating a Simple Formula**

- 1) Create a blank workbook. 
- 2) Click in cell **A1** and enter the following.

=5+5

- 3) Press **[Enter]** to complete the formula. Excel will calculate the result.

**Exercise 2. Using Cell Referencing**

- 1) Enter the number 5 in cell **B1**.
- 2) Enter the number 10 in cell **B2**.
- 3) Enter the following formula in cell **B3**.

=b1+b2

- 4) Press **[Enter]** to complete the formula. Excel will calculate the result.
- 5) Change the number in cell **B2** and press **[Enter]**. The result of the formula will be re-calculated (if it doesn't recalculate on its own you can press **[F9]** to force recalculation).


**Exercise 3. Using the Mouse for Cell Referencing**

Creation of a formula can be made easier by using the mouse to create cell references.

- 1) Enter the number 10 in cell **C1**.
- 2) Enter the number 20 in cell **C2**.
- 3) Click in cell **C3** and type a = sign.
- 4) Click on cell **C1**. The reference for that cell will appear in the formula.
- 5) Type a + sign.
- 6) Click on cell **C2**. The reference for that cell will appear in the formula.
- 7) Check that the formula reads =C1+C2 and press **[Enter]**.

### Exercise 4. Editing a Formula

- 1) Select cell **C3** and press **[F2]** to edit the formula.



Notice that the cell references in the formula have become coloured. The cells the references refer to have a border around them that is the same colour as the reference. These coloured borders can be moved as an easy way to edit a formula.

- 2) Move your mouse to the edge of the blue border around cell **C1**.
- 3) Drag the border to move it to cell **B1**. This will change the cell reference in the formula.
- 4) Press **[Esc]** to cancel the changes to in the formula.
- 5) Close the workbook without saving the changes.

### Exercise 5. Use of Brackets

- 1) Open the workbook called *Formulas*.
- 2) Click on cell **B11** and enter the current date by pressing **[Ctrl] [;]** and then pressing **[Enter]**.
- 3) Click on cell **B12** and enter your date of birth.

We will create a formula in cell **B13** that calculates your age by taking your date of birth away from the current date.

- 4) Click in cell **B13** and enter the following formula.

`=B11-B12`

The result of the formula will show your age just as it should. Unfortunately, it's showing you your age in days instead of in years. We can correct this by dividing the result by the number of days in a year.

- 5) Edit the formula so that it looks like the one below and press **[Enter]**.

`=B11-B12/365`

This still won't give you the correct answer because Excel will calculate division before it will calculate subtraction (remember the order of operations). We need to tell Excel to calculate the subtraction first and *then* divide the result by 365. This is where the brackets come in.

- 6) Edit the formula again to look like the one below and press **[Enter]**.

`=(B11-B12)/365`

That's more like it. The brackets tell Excel to calculate that part of the formula first.

## Functions

Functions are a special kind of formula that can simplify complex calculations. The following exercises will demonstrate the use of functions.

### Exercise 6. Adding Several Numbers Together

- 1) Make sure the Formulas workbook is still open.
- 2) Click in cell **B9**.
- 3) Begin your formula with a = sign and then click on cell **B2**.
- 4) Type a + sign and click on cell **B3**.
- 5) Complete the formula so that it looks like the one below.

=B2+B3+B4+B5+B6+B7+B8

You will get the correct answer, but it's a long formula. Adding together 50 or 100 cells like this would be a tedious process.

Functions can be used to take the hard work out of many types of calculations in Excel. Functions all follow the same format. I.e. = sign, name of the function, information to be calculated in brackets.

=NameOfFunction(information to be calculated)

There are hundreds of functions built in to Excel and custom functions can be created. The most commonly used function is the Sum function.

### Exercise 7. Sum Function

- 1) Click in cell **C9**.
- 2) Type =sum(C2:C8) and press **[Enter]**.
- 3) Select cells **C2** to **C8**.

This will tell Excel to add up the sum of all the cells from **C2** to **C8**.

- 4) Click in cell **D9**.
- 5) Type =sum(  
6) Select cells **D2** to **D8**.
- 7) You can type the right bracket, but if you don't Excel will put it in for you. Press **[Enter]** to complete the function. It should look like the one below.

=SUM(D2:D8)

Note: If you want to add together more than one group of cells, you can separate each cell range with a comma.

E.g. =SUM(C2:C8,E2:E8,G2:G8)


This would add cells **C2** to **C8** and **E2** to **E8** and **G2** to **G8**

### Exercise 8. Using the AutoSum Tool

Since the Sum function is used so often, a special tool has been provided to make it easy to use.

- 1) Click in cell **E9**.
- 2) Click the AutoSum icon on the toolbar.  $\Sigma$  ▾

Excel will create a Sum function referring to the cells above. Excel will assume the cells above are the ones to be added together. These cells will remain selected in case you would rather select a different group of cells. If there are numbers in the cells above, the cells to the left will be used. Otherwise the nearest group of cells will be selected.



- 3) Press **[Enter]** to confirm that the correct cells are selected and complete the function.

The function should look like the one below.

`=SUM(E2:E8)`

You can also use the AutoSum tool by selecting the cells to be added first.

- 4) Select cells **F2 to F8**.
- 5) Click the AutoSum icon.  $\Sigma$  ▾


A Sum function will be automatically created based on the cells you selected.

The AutoSum icon can also be used to create other common functions such as Average and Count.

- 6) Click in cell **G9**.
- 7) Click the arrow next to the AutoSum icon.

A list of common functions will be displayed.


- 8) Click on **Average**.



An Average function will be created in the selected cell. Notice that it is written the same as a Sum function.

- 9) Check that the function reads `=AVERAGE(G2:G8)` and press **[Enter]** to complete the function.
- 10) Save the changes to the workbook and then close it.

**Exercise 9. Using Functions in a Table**



- 1) Open the Grades workbook (This was created in the [Getting Started](#) exercises).
- 2) Click in cell **F6**.
- 3) Click the AutoSum icon. 
- 4) Make sure cells **B6 to E6** are selected and press **[Enter]** to complete the function.

15	12	14	17	=SUM(B6:E6)
18	14	17	16	[SUM(number1, [number2], ...)]

Year Total	58

- 5) Use the fill handle to copy the function down through to cell **F16**.

When a formula is copied to other cells, the cell references should change for each cell.

- 6) Click in cell **B15**.
- 7) Click the arrow next to the AutoSum icon and click on Average. 
- 8) Make sure cells **B6 to E6** are selected and press **[Enter]** to complete the function.
- 9) Use the fill handle to copy the formula across to cell **F15**.
- 10) Click in cell **B16**.
- 11) Click the arrow next to the **AutoSum** icon and click on **Max**. 

Notice that the function has selected all of the cells above including the average cell which we don't want selected.

- 12) While these cells are still highlighted, select cells **B6 to B14** and press **[Enter]**. The formula in cell **B16** should be =MAX(B6:B14).
- 13) Use the fill handle to copy the formula across to cell **F16**.
- 14) Try completing the Lowest mark cells yourself by using the **Min** function.
- 15) Use a count function in cell **B18** to show the number of students.

The completed table should look like the one below.

Joe Bloggs Eastern Goldfields Senior High School					
15					
24/11/2003					
Student	Term 1	Term 2	Term 3	Term 4	Year Total
Jean Picard	15	12	14	17	58
John Archer	18	14	17	16	65
James Kirk	23	22	19	21	85
Chris Pike	8	11	7	6	32
Jeff Sinclair	19	19	18	14	70
Katheryn Ja	13	12	10	12	47
Ben Cisco	16	22	20	19	77
John Sherric	22	20	24	21	87
Lita Alexanc	23	21	28	21	93
Class Avera	17.44444	17	17.44444	16.33333	68.22222
Highest Mar	23	22	28	21	93
Lowest Mar	8	11	7	6	32
Number of S	9				

**Exercise 10. Using an If Function**

If functions can be useful if you want an excel formula to select from 2 different answers based on criteria you specify. In our grades example, we can use an If function to say a student has passed if their mark was over 50 and fail if their mark wasn't over 50. If functions have 3 sections with a comma between each section as shown below.

=If(condition to test, answer if condition is true, answer if condition is false)

- 1) Click in cell G5 and enter the heading *Pass/Fail*.
- 2) Enter the following If function.

=If(F6>=50,"Pass","Fail")

The first part of the formula checks to see if the number in cell F6 is greater than 50 or equal to 50. The = sign is important. If it was left out then someone who scored exactly 50 wouldn't be included. The second section says that if this test condition is true, the text *Pass* will be displayed for the answer. Note that where text is used in a formula, it must be enclosed in quotation marks. The third section specifies the answer (*Fail*) if the test condition is not true.

- 3) When the If function has been entered, use the fill handle to copy the formula down to cell G14. There should be 7 passes and 2 fails.

## Absolute Cell Referencing

In the previous exercise, you copied formulas to different cells. As the formulas were copied, the cells referred to in each formula changed to suit the new location. This is because formulas normally use *relative* referencing.

	A	B	C
1	12	14	=A1*B1
2	13	15	=A2*B2

In the example above, the formula in **C1** multiplies **A1** by **B1**. When the formula is copied downward, it changes to multiplying **A2** by **B2**. This is because Excel is not keeping track of exact cell references. It is only seeing the cells in the formula are the two cells to the left. When the formula is copied down, it is still copying the two cells to the left.

Relative references are fine for most formulas but sometimes there are situations where you want a formula to refer to a specific cell even when you copy the formula to another location. In these situations you need to use *absolute* references.

- 1) Open the Fqto.xls workbook.
- 2) Click in cell **I2** and enter 30%.
- 3) Click in cell **H2** and enter =G2\*I2.

You should get the correct answer for this cell.

- 4) Use the fill handle to copy this formula down to cell **H8**.

All of these cells will be filled with 0s. When the formula is copied down, it will still be referring to the cell to the right of the formula and all of the cells the formula has been copied to have nothing to the right. We will need to edit the original formula to make sure that it keeps on referring to cell **I2**, even when the formula is copied to other locations.

- 5) Click in cell **H2** and press **F2** to edit the formula.
- 6) Click on the part of the formula that refers to **I2**.
- 7) Press **[F4]**.

Dollar signs will appear in the cell reference (**\$I\$2**). These dollar signs mean that the formula will still refer to column **I** and row **2** even when it is copied to another location. You can type the dollar signs yourself but the **F4** shortcut will cycle between the different kinds of references.

- |        |   |
|--------|---|
| I2     | Relative reference – row and column references will both change.  |
| \$I2   | Mixed reference – Column reference stays constant, row reference will change.                                       |
| I\$2   | Mixed reference – Row reference stays constant, column reference will change.                                       |
| \$I\$2 | Absolute reference – Row and column reference will both stay constant if the formula is copied to another location. |

- 8) Make sure the formula in **H2** reads =G2\*\$I\$2 and press **[Enter]**.
- 9) Use the fill handle to copy the formula down to cell **H8**. Each formula should now be still correctly referring to cell **I2**. Save and close the workbook.



## Lookup Functions

Lookup functions are a bit like an IF function. They allow you to create a formula that will choose from several different answers based on information in a table. We will use a vertical lookup function to create a formula which checks a student's mark and shows what their grade will be.

### Exercise 11. Vertical Lookup

- 1) Open the Grades workbook if it is not already open.

The first thing we will need to do is create the table that will contain the grade cut-off points.

- 2) Click in cell **K6** and enter a zero.
- 3) Complete the information in cells **K6** to **L10** as shown to the right.
- 4) Click in cell **H5** and enter the text *Grade*.
- 5) Click in cell **H6**. This is where we want the first grade to go.

K	L
	0 F
	35 D
	50 C
	65 B
	80 A

Like an **If** function, a **Lookup** function has different parts. The sections of a lookup function are described below.

Lookup value	The cell the function is checking, in this case the student mark
Table array	The cells the answers are coming from - our table of grade cut-offs
Column index	The column in the array that the answer will come from. In our case, the function will match the student grade with a number in the first column of the array, then it will check the second column to find the matching grade, so the second column is the column index.
Range lookup	If the function doesn't find an exact match in the table array, putting true in this section will mean that it will use the closest match instead.

- 6) Enter the following formula. =**VLOOKUP**(
- 7) Click on cell **F6** (the cell with the first student's mark).
- 8) Type a comma to end the first section and then select the grades cut-off table (which should be cells **K6** to **L10**).
- 9) Press **F4** to make sure this section uses absolute references (**\$K\$6:\$L\$10**). When we copy the function down for the other students we want to make sure it is still correctly referring to the cells with the grade cut-off marks.
- 10) Type another comma and then the number 2.
- 11) Type another comma and then type *true*.
- 12) Type a closing bracket and then press **[Enter]** to complete the formula. The completed formula should be =**VLOOKUP**(**F6,\$K\$6:\$L\$10,2,TRUE**).
- 13) Use the fill handle to copy the function down to cell **H14** so that every student has a grade.

Year Total	Pass/Fail	Grade
58	Pass	C
65	Pass	B
85	Pass	A
32	Fail	F
70	Pass	B
47	Fail	D
77	Pass	B
87	Pass	A
93	Pass	A

If your lookup table goes horizontally instead of vertically then you can use a horizontal lookup function. It works exactly the same except that you type **Hlookup** instead of **Vlookup**.

- 14) Save and close the file.